

定理証明支援系 CoQ での 理論的なリーゾニング

集中講義 千葉大学大学院

アフエルト レナルド

産業技術総合研究所

2017 年 01 月 25-26 日

目的

- ▶ Coq [CDT16] は型理論の実装である. 型理論で命題論理と述語論理を表現しやすい.
- ▶ Coq でタクティック [CDT16, Chapitre 8] を用いてリーゾニングを行う. タクティックは自然演繹のルールに近い.
- ▶ Coq のタクティックを用いて, ([DNR03] による) 自然演繹のルールを説明する.
- ▶ Coq は型付きラムダ計算であることは重要である: 型は理論式であり, 項は証明である.
- ▶ 式が成り立つというのは, その型を持つ項があるということである. Coq は **Curry-Howard** 同型対応の一例である.

アウトライン

CoQ の概要

CoQ での命題論理

公理と含意

論理積と論理和

矛盾と否定


CoQ での述語論理

補足

定理証明支援系 Coq

- ▶ INRIA で T. Coquand と G. Huet が 1984 年に開発を始めた
- ▶ 基本的に, Gallina というラムダ計算である
 - ▶ Calculus of Inductive Constructions (CIC) [CP90, PM92]
 - ▶ Calculus of Constructions の拡張 [CH84, CH85, CH86, CH88]
- ▶ たくさん使われている
 - ▶ 世界中で, 研究と教育で
 - ▶ 賞: ACM SIGPLAN Programming Languages Software 2013 award, ACM Software System 2013 award
 - ▶ 産業への応用の試み (CC EAL7 認証)
- ▶ 成功例: 四色定理 [Gon05, Gon08], C のコンパイラ [Ler09, BL09], 奇数位数定理 [GAA⁺13], 等

Coq システムの概要



→: コマンド言語を用いて, 補題/データ構造の追加 **Vernacular**

-->: Gallina を用いて, 項の直接的記述

~>: 型/証明検査の失敗の際の情報/エラーメッセージ

→▷: 実際に, **タクティク**を用いて, 項を間接に記述する

—○: 標準ライブラリはデータ構造, タクティクや証明済みの補題を提供する

Coqのインタフェース

入力

コマンドとタクティク

```
Require Import  
Classical.  
  
Goal forall (P Q : Prop),  
  (P -> Q) -> P -> Q.  
Proof.  
intros P Q.
```

⋮
証明のスクリプト
⋮

出力

ゴール

$P : \text{Prop}$
 $Q : \text{Prop}$ } ローカルコンテキスト (Γ)

===== 水平線

$(P \rightarrow Q) \rightarrow P \rightarrow Q$
 { }
 トップ ゴール
 { }
 仮定のスタック

エラーメッセージ,
検索の結果, 等

Coq の出力をシークエントとして読む

- $$\begin{array}{l} p : P \\ q : Q \\ \hline R \end{array}$$
- ▶ $P, Q \vdash R$ として読む
 - ▶ P, Q, R は命題である
 - ▶ 仮定 (P, Q) の証明 (p, q) は明確である
 - ▶ 結論 R の (構築中の) 証明は **Show Proof** コマンドで表示できる

アウトライン

CoQ の概要

CoQ での命題論理

公理と含意

論理積と論理和

矛盾と否定

CoQ での述語論理

補足

自然演繹ルール「公理」

- ▶ 「証明があれば, 命題を証明できる」
- ▶ タクティク **exact** によって実装されている
 - ▶ 型検査しか呼ばないので, もっとも基本的なタクティクである

$$\frac{}{\Gamma, A \vdash A} \text{ax}$$

A : Prop

a : A

=====

A

→**exact** a. → No more subgoals.

含意 (\rightarrow) の導入

- ▶ 「 $A \rightarrow B$ を証明するために, A を仮定してから B を証明するのは十分である」
- ▶ タクティク `intros/revert` を用いて, \vdash の前/後に仮定を動かすだけ
 - ▶ ローカルコンテキストの仮定の証明に, 名前を付けなければならない

$$\frac{\Gamma, A \vdash B}{\Gamma \vdash A \rightarrow B} \rightarrow_i$$

$A, B : \text{Prop}$
=====
 $A \rightarrow B$

\rightarrow `intros` $a.^1 \rightarrow$
 \leftarrow `revert` $a.^2 \leftarrow$

$A, B : \text{Prop}$
 $a : A$
=====
 B

¹一つのパラメータだけなら, `intro` も使える

²`revert` はローカルコンテキストから仮定を消す. `generalize` はコピーを行う.

含意 (\rightarrow) の除去

- ▶ 「 $A \rightarrow B$ と A を知ると, B を推論できる」 (modus ponens)
- ▶ 含意の除去は関数の適用として理解する
 - ▶ もし ab は $A \rightarrow B$ の証明であり, a は A の証明であれば, ab a は B の証明である
- ▶ 一つのサブゴールしか生成しない **apply** タクティクを使用する³

$$\frac{\Gamma \vdash A \rightarrow B \quad \Gamma \vdash A}{\Gamma \vdash B} \rightarrow_e$$

$A, B : \text{Prop}$

$ab : A \rightarrow B$

$a : A$

=====

B

\rightarrow **apply** $ab.$ \rightarrow

$A : \text{Prop}$

$ab : A \rightarrow B$

$a : A$

=====

A

³**cut** は \rightarrow_e にもっと近いが, 不便である

証明の例 (1/7)

ゴールの記述

$$\vdash (A \rightarrow B \rightarrow C) \rightarrow (A \rightarrow B) \rightarrow A \rightarrow C$$

```
Lemma hilbertS (A B C : Prop) :  
  (A -> B -> C) -> (A -> B) -> A -> C.  
Proof.
```

証明の例 (2/7)

一番目の含意の導入

$$\frac{\overbrace{A \rightarrow B \rightarrow C}^{abc} \vdash (A \rightarrow B) \rightarrow A \rightarrow C}{\vdash (A \rightarrow B \rightarrow C) \rightarrow (A \rightarrow B) \rightarrow A \rightarrow C} \rightarrow_i$$

```
Lemma hilbertS (A B C : Prop) :  
  (A -> B -> C) -> (A -> B) -> A -> C.  
Proof.  
intros abc.
```

証明の例 (3/7)

三つの含意の導入

$$\frac{\frac{\frac{\overbrace{A \rightarrow B \rightarrow C}^{abc}, \overbrace{A \rightarrow B}^{ab}, \overbrace{A}^a \vdash C}{A \rightarrow B \rightarrow C, A \rightarrow B \vdash A \rightarrow C} \rightarrow_i}{A \rightarrow B \rightarrow C \vdash (A \rightarrow B) \rightarrow A \rightarrow C} \rightarrow_i}{\vdash (A \rightarrow B \rightarrow C) \rightarrow (A \rightarrow B) \rightarrow A \rightarrow C} \rightarrow_i$$

```
Lemma hilbertS (A B C : Prop) :  
  (A -> B -> C) -> (A -> B) -> A -> C.
```

```
Proof.
```

```
intros abc ab a.
```

証明の例 (4/7)

含意の一番目の除去

$$\frac{A \rightarrow B \rightarrow C, A \rightarrow B, A \vdash A \quad A \rightarrow B \rightarrow C, A \rightarrow B, A \vdash B}{\text{abc} \quad \frac{\frac{A \rightarrow B \rightarrow C, A \rightarrow B, A \vdash C}{A \rightarrow B \rightarrow C, A \rightarrow B \vdash A \rightarrow C} \rightarrow_i \quad \frac{A \rightarrow B \rightarrow C \vdash (A \rightarrow B) \rightarrow A \rightarrow C}{\vdash (A \rightarrow B \rightarrow C) \rightarrow (A \rightarrow B) \rightarrow A \rightarrow C} \rightarrow_i} \rightarrow_e$$

```
Lemma hilbertS (A B C : Prop) :  
  (A -> B -> C) -> (A -> B) -> A -> C.
```

Proof.

```
intros abc ab a.
```

```
apply abc.
```

($A \rightarrow B \rightarrow C$ の結論と C をマッチし、二つのサブゴール A と B を生成する)

証明の例 (5/7)

「公理」ルールを適用

$$\frac{\frac{\frac{\frac{\frac{}{A \rightarrow B \rightarrow C, A \rightarrow B, \overbrace{A \vdash A}^a}{}{A \rightarrow B \rightarrow C, A \rightarrow B, A \vdash B}}{A \rightarrow B \rightarrow C, A \rightarrow B, A \vdash C} \rightarrow_i}{A \rightarrow B \rightarrow C, A \rightarrow B \vdash A \rightarrow C} \rightarrow_i}{A \rightarrow B \rightarrow C \vdash (A \rightarrow B) \rightarrow A \rightarrow C} \rightarrow_i}{\vdash (A \rightarrow B \rightarrow C) \rightarrow (A \rightarrow B) \rightarrow A \rightarrow C} \rightarrow_e \quad \text{ax}$$

Lemma hilbertS (A B C : Prop) :
(A -> B -> C) -> (A -> B) -> A -> C.

Proof.

intros abc ab a.

apply abc.

- **exact** a.

含意の二番目の除去

```
Lemma hilbertS (A B C : Prop) :
  (A -> B -> C) -> (A -> B) -> A -> C.
```

```
intros abc ab a.
```

```
apply abc.
```

- exact a.

- apply ab.

◀ ◻ ▶ ◀ ◻ ▶ ◀ ≡ ▶ ◀ ≡ ▶ ≡

証明の例 (7/7)

公理と Qed

$$\frac{\frac{\frac{A \rightarrow B \rightarrow C, A \rightarrow B, A \vdash A}{A \rightarrow B \rightarrow C, A \rightarrow B, A \vdash A} \text{ ax} \quad \frac{\frac{A \rightarrow B \rightarrow C, A \rightarrow B, \overbrace{A \vdash A}^a}{A \rightarrow B \rightarrow C, A \rightarrow B, A \vdash B} \rightarrow_e}{A \rightarrow B \rightarrow C, A \rightarrow B, A \vdash B} \rightarrow_e}{A \rightarrow B \rightarrow C, A \rightarrow B, A \vdash C} \rightarrow_i}{A \rightarrow B \rightarrow C, A \rightarrow B \vdash A \rightarrow C} \rightarrow_i}{A \rightarrow B \rightarrow C \vdash (A \rightarrow B) \rightarrow A \rightarrow C} \rightarrow_i}{\vdash (A \rightarrow B \rightarrow C) \rightarrow (A \rightarrow B) \rightarrow A \rightarrow C} \rightarrow_i$$

```
Lemma hilbertS (A B C : Prop) :  
  (A -> B -> C) -> (A -> B) -> A -> C.
```

```
Proof.
```

```
intros abc ab a.
```

```
apply abc.
```

```
- exact a.
```

```
- apply ab.
```

```
  exact a.
```

```
Qed.
```

Qed (Quod Erat Demonstrandum) を hilbertS の識別子を用いて, 証明を記録する

証明と証明のスキプトの違い

- ▶ 実は, 書いたものは, 証明のスキプト
- ▶ 証明自体は次のように表示できる:

```
Lemma hilbertS (A B C : Prop) :  
  (A -> B -> C) -> (A -> B) -> A -> C.
```

```
Proof.
```

```
intros abc ab a.
```

```
apply abc.
```

```
- exact a.
```

```
- apply ab.
```

```
  exact a.
```

```
Show Proof.
```

```
(fun (A B C : Prop) (abc : A -> B -> C) (ab : A -> B) (a : A) =>  
  abc a (ab a))
```

- ▶ タクティクはラムダ計算の項を書くための間接な方法だけである. ただし, その項は証明として見る.
- ▶ Coq は Curry-Howard 同型対応の実装の一例である

証明項の例

Hilbert の S 公理

- 途中のゴールの裏に隠れている (構築中の) 証明項をすべて表示する:

$$\begin{array}{c} \frac{\frac{\Gamma \vdash a : A}{\Gamma^4 \vdash abc\ a\ (aba) : C} \quad \frac{\frac{\Gamma \vdash a : A}{\Gamma \vdash aba : B}}{\Gamma^4 \vdash abc\ a\ (aba) : C}}{\frac{\frac{abc : A \rightarrow B \rightarrow C, ab : A \rightarrow B \vdash \lambda a : A. abc\ a\ (aba) : A \rightarrow C}{abc : A \rightarrow B \rightarrow C \vdash \lambda ab : A \rightarrow B. \lambda a : A. abc\ a\ (aba) : (A \rightarrow B) \rightarrow A \rightarrow C}}{\vdash \lambda abc : A \rightarrow B \rightarrow C. \lambda ab : A \rightarrow B. \lambda a : A. abc\ a\ (aba) : (A \rightarrow B \rightarrow C) \rightarrow (A \rightarrow B) \rightarrow A \rightarrow C}} \end{array}$$

- どうして Coq のインタフェースが証明項を毎回表示しないことが分かるでしょう...

⁴ $\Gamma = abc : A \rightarrow B \rightarrow C, ab : A \rightarrow B, a : A$

アウトライン

CoQ の概要

CoQ での命題論理

公理と含意

論理積と論理和

矛盾と否定

CoQ での述語論理

補足

論理積 (\wedge) の導入

- ▶ 「 $A \wedge B$ を証明するために, A と B を証明するのは十分である」⁵
 - ▶ 証明木の中に枝を追加すること
- ▶ その論法はタクティク `split` によって実装されている

$$\frac{\Gamma \vdash A \quad \Gamma \vdash B}{\Gamma \vdash A \wedge B} \wedge_i$$

$A, B : \text{Prop}$

$a : A$

$b : B$

=====

$A \wedge B$

$\rightarrow \text{split} \rightarrow$

$A, B : \text{Prop}$

$a : A$

$b : B$

=====

A

$A, B : \text{Prop}$

$a : A$

$b : B$

=====

B

⁵はい, 当たり前のように聞こえる

論理積の (\wedge) の除去

- ▶ 「 A (resp. B) を証明するために, $A \wedge B$ を証明すれば十分である」 (sic)
 - ▶ アイデア: 論理積を特化できる
- ▶ タクティク **exact** と proj1/proj2 補題の組み合わせ

$$\frac{\Gamma \vdash A \wedge B}{\Gamma \vdash A} \wedge_e^g \quad \left(\text{resp. } \frac{\Gamma \vdash A \wedge B}{\Gamma \vdash B} \wedge_e^d \right)$$

```
A, B : Prop
ab : A /\ B
=====
A      →exact (proj1 ab)6→ No more subgoals.
```

⁶resp. proj2

どうやって論理積 (\wedge) は定義されている?

論理積は証明のペアの型として定義されている:

$\text{Inductive and (A B : Prop) : Prop :=}$
構成子 \longrightarrow $\text{conj : A } \rightarrow \text{ B } \rightarrow \text{ A } \wedge \text{ B.}$
型 パラメータ ソート
構成子の型

- ▶ A と B は命題 (Prop) であれば, $A \wedge B$ も命題である
- ▶ $A \wedge B$ は `and A B` のための記法である
- ▶ a は A 証明であり, b は B の証明であれば, `conj a b` は $A \wedge B$ の証明である

論理積 (\wedge) の使用に当たるテクニカルな話

- ▶ タクティク `split` は関数として見た構成子 `conj` の適用に当たる:

`split` \approx `apply conj`

- ▶ `proj1/proj2` は標準の補題である:

Theorem `proj1` : $A \wedge B \rightarrow A$.

Theorem `proj2` : $A \wedge B \rightarrow B$.

- ▶ 実際に, `proj1` と `proj2` を使うより, タクティク `destruct` の方が便利である:

`A, B : Prop`

`ab : A \wedge B`

=====

`A`

\rightarrow `destruct ab as [a b]` \rightarrow

`A, B : Prop`

`a : A`

`b : B`

=====

`A`

論理積を使う証明項の例

タクティク **destruct** は Gallina (CoQ の言語) でのパターンマッチの間接な方法である:

```
Lemma myproj1 (A B : Prop) : A /\ B -> A.  
Proof.  
  exact (fun ab : A /\ B =>  
    match ab with conj a b => a end).  
Qed.
```

論理和 (\vee) の導入

- ▶ 「 $A \vee B$ を証明するために、 A または B を証明するのは十分である」
 - ▶ アイデア: このルールを用いて、ゴールを特化する
 - ▶ \wedge_e^g / \wedge_e^d と比較
- ▶ この論法はタクティック **left** と **right** によって実装されている

$$\frac{\Gamma \vdash A}{\Gamma \vdash A \vee B} \vee_i^g \quad \left(\text{resp.} \quad \frac{\Gamma \vdash B}{\Gamma \vdash A \vee B} \vee_i^d \right)$$

$A, B : \text{Prop}$

$a : A$

=====

$A \ \backslash / \ B$

$\rightarrow \text{left}^7 \rightarrow$

$A, B : \text{Prop}$

$a : A$

=====

A

⁷resp. **right**

論理和 (\vee) の除去

- ▶ 論理和を用いて、場合分けを行う
 - ▶ 証明木の中に、枝を追加する
 - ▶ \wedge_i と比較
- ▶ タクティク `destruct` を使う

$$\frac{\Gamma \vdash A \vee B \quad \Gamma, A \vdash C \quad \Gamma, B \vdash C}{\Gamma \vdash C} \vee_e$$

$A, B, C : \text{Prop}$

$ab : A \vee B$

=====

C

$\rightarrow \text{destruct } ab \text{ as } [a|b] \rightarrow$

$A, B : \text{Prop}$

$a : A$

=====

C

$A, B : \text{Prop}$

$b : B$

=====

C

($[a|b]$, 論理積の場合の $[a \ b]$ と違う—[slide 25](#))

どうやって論理和は定義されていますか?

論理和は二つの構成子を持つ型として定義されている:

```
Inductive 型 or パラメータ (A B : Prop) : ソート Prop :=  
構成子 → | or_introl : A -> A \/  
          | or_intror : B -> A \/  
                                     構成子の型 B
```

- ▶ A と B は命題 (Prop) であれば, A \/
B も命題である
- ▶ A \/
B は or A B のための記法
- ▶ a は A の証明であると, or_introl B a は A \/
B の証明になる
- ▶ b は B の証明であると, or_intror A b は A \/
B の証明になる

論理和 (V) の使用に当たるテクニカルな話

タクティク `left` は関数として見る構成子 `or_introl` の適用である:

$$\text{left} \approx \text{apply } \text{or_introl}$$

タクティク `right` は関数として見る構成子 `or_intror` の適用である:

$$\text{right} \approx \text{apply } \text{or_intror}$$

論理和を使う証明項の一例

論理和の除去のルール

$$\text{メモ: } \frac{\Gamma \vdash A \vee B \quad \Gamma, A \vdash C \quad \Gamma, B \vdash C}{\Gamma \vdash C} \vee_e$$

```
Lemma or_elim (A B C : Prop) :  
  (A -> C) -> (B -> C) -> A \/ B -> C.  
Proof.
```

論理和を使う証明項の一例

タクティクを用いた証明

```
Lemma or_elim_tactique (A B C : Prop) :  
  (A -> C) -> (B -> C) -> A \/ B -> C.  
Proof.  
  intros ac bc.  
  destruct 1 as [a | b].  
- apply ac.  
  exact a.  
- apply bc.  
  exact b.  
Qed.
```


論理和を使う証明項の一例

Gallina 関数を用いた証明

```
Lemma or_elim_gallina (A B C : Prop) :  
  (A -> C) -> (B -> C) -> A \/ B -> C.  
Proof.  
exact (fun (ac : A -> C) (bc : B -> C) (ab : A \/ B) =>  
match ab with  
| or_introl a => ac a  
| or_intror b => bc b  
end).  
Qed.
```

実際に、両方のスクリプトは 同じ証明 を構築する.

アウトライン

CoQ の概要

CoQ での命題論理

公理と含意

論理積と論理和

矛盾と否定

CoQ での述語論理

補足

Coqでの矛盾と否定

- ▶ 矛盾 (紙上の記法: \perp) は Coq で `False` と書く
 - ▶ `False` は基本的な概念でない. 定義されている ([slide 39](#))
- ▶ A の否定 (紙上の記法: $\neg A$) は Coq で $\sim A$ (「tilda A 」と読む) と書く
 - ▶ \sim も基本的な結合子ではない. 定義されている:
 - ▶ 紙上, $\neg A$ は $A \rightarrow \perp$ として定義されている
 - ▶ Coq で, $A \rightarrow \perp$ は $A \rightarrow \text{False}$ となる
 - ▶ 見えやすくするために, $A \rightarrow \text{False}$ の代わりに, $\sim A$ と書く

否定 (\neg) の導入と除去

メモ [DNR03]:

$$\frac{\Gamma, A \vdash \perp}{\Gamma \vdash \neg A} \neg_i \quad \frac{\Gamma \vdash \neg A \quad \Gamma \vdash A}{\Gamma \vdash \perp} \neg_e$$

上記のルールで, $\neg A$ の代わりに, その定義 $A \rightarrow \perp$ と書く:

$$\frac{\Gamma, A \vdash \perp}{\Gamma \vdash A \rightarrow \perp} \neg_i \quad \frac{\Gamma \vdash A \rightarrow \perp \quad \Gamma \vdash A}{\Gamma \vdash \perp} \neg_e$$

そうすると, 否定のルールは含意のルールのインスタンスであることが分かる (下記のもの B の代わりに, \perp と書いただけ):

$$\frac{\Gamma, A \vdash B}{\Gamma \vdash A \rightarrow B} \rightarrow_i \quad \frac{\Gamma \vdash A \rightarrow B \quad \Gamma \vdash A}{\Gamma \vdash B} \rightarrow_e$$

従って, 否定のために, 新しいタクティクを加える必要はない

Coqでの否定 (\neg) の導入

「 $\neg A$ をするために, A が成り立つコンテキストで \perp を証明するのは十分である」

$$\frac{\Gamma, A \vdash \perp}{\Gamma \vdash \neg A} \neg_i$$

A : Prop

===== → intros a. →

~ A

A : Prop

a : A

=====

False

Coq での否定 (\neg) の除去

「 \perp を証明するために, A と $\neg A$ が成り立つ A を見つかるのは十分である」

$$\frac{\Gamma \vdash \neg A \quad \Gamma \vdash A}{\Gamma \vdash \perp} \neg_e$$

```
A : Prop
na : ~ A
a : A      → apply na.8→ No more subgoals.
=====
False
```

⁸または: **exact** (na a)

CoQ で、どうやって矛盾 (\perp) が定義されている?

- ▶ \perp は証明の作れない命題である
- ▶ CoQ で, \perp は要素のない型として定義されている
- ▶ 具体的に, `False` は構成子のない帰納的型として定義する.
次のコマンドで実現する:

```
Inductive Type False : Sort Prop := .
```

矛盾 (\perp) の除去のルール

- ▶ 実際に, コンテキストで \perp があれば, 除去するだけで, ゴールを証明する
- ▶ *ex falso quodlibet*: 矛盾から, 何でも証明できる

$$\frac{\Gamma \vdash \perp}{\Gamma \vdash A} \perp_e$$

```
A : Prop
abs : False
===== →destruct abs.→ No more subgoals.
A
```


CoQでの古典論理

- ▶ CoQは**構成的論理**: 古典論理の性質を使えないように定義されている
 - ▶ 例: 排中律 (「命題は成り立つまたは成り立たない」)
- ▶ ただし, 古典論理を導入しても, 矛盾を導かない
 - ▶ CoQで古典論理は**公理** (証明のない補題) として用意されている

「証明できる」古典論理の論法の低:

「もし $\neg A$ から \perp を証明できれば, A は成り立つ」

```
Require Import Classical.
```

```
Lemma bottom_c (A : Prop) : ((~A) -> False) -> A.  
Proof. ... Qed.
```

古典論理による矛盾

古典論理の矛盾のルールは前のスライド ([slide 41](#)) の `bottom_c` 補題を用いて証明できる:

$$\frac{\Gamma, \neg A \vdash \perp}{\Gamma \vdash A} \perp_c$$

```
A : Prop
=====
A
```

→

```
apply bottom_c.
intros na.
```

→

```
A : Prop
na : ~ A
=====
False
```

自然演繹の「Weakening」ルール

全てのルールを列挙するように...

- ▶ 不要な仮定は出てくることがある
- ▶ この論法はタクティク `clear` を用いて実現できる

$$\frac{\Gamma \vdash A}{\Gamma, B \vdash A} \text{ weak}$$

A, B : Prop

a : A

b : B

=====

A

→ `clear` b. →

A, B : Prop

a : A

=====

A

アウトライン

CoQ の概要

CoQ での命題論理

公理と含意

論理積と論理和

矛盾と否定

CoQ での述語論理

補足

全称記号の導入

- ▶ 「 $\forall x.A$ を証明するために、任意の x に対して A を証明するのは十分である」
- ▶ 楽なところ:
 - ▶ $\forall(x:X), A$ は $X \rightarrow A$ のただの一般化 (特に、同じタクティク)
 - ▶ CoQ が識別子の管理を自動的に行う

$$\frac{\Gamma \vdash A \quad x \text{ は } \Gamma \text{ の式の自由変数ではない}}{\Gamma \vdash \forall x.A} \forall_i$$

```
X : Type
A : X -> Prop
x : X
=====
forall x : X, A x
```

```
→intros x0.→
←revert x0.9←
```

```
X : Type
A : X -> Prop
x : X
x0 : X
=====
A x0
```

⁹`generalize` `x0` は新しい変数 `x1` を作成する

全称記号の除去

- ▶ 全称記号の除去は関数の適用として理解する
- ▶ 含意の除去に似ている (特に, 同じタクティク)

$$\frac{\Gamma \vdash \forall x. A}{\Gamma \vdash A[x := t]} \forall_e$$

X : Type

A : X -> Prop

Ax : forall x : X, A x

t : X

=====

A t

→**apply** Ax.¹⁰→

No more
subgoals.

¹⁰または **exact** (Ax t)

存在記号の導入

- ▶ 「 $\exists x.A$ を証明するために, $A\ t$ が成り立つように witness t を見つけるのは十分である」
- ▶ この論法は **exists** タクティクとして実装されている

$$\frac{\Gamma \vdash A[x := t]}{\Gamma \vdash \exists x.A} \exists_i$$

```
X : Type
x : X
A : X -> Prop
t : X
At : A t
=====
exists x0 : X, A x0
```

\rightarrow exists t \rightarrow

```
X : Type
x : X
A : X -> Prop
t : X
At : A t
=====
A t
```

存在記号の除去

- ▶ $\exists x.A$ は witness t と証明 $A\ t$ のペアとして見る
- ▶ タクティク **destruct** は witness とその証明を明確にする
 - ▶ Coq が問題を起こさない識別子を選ばれることを確認する

$$\frac{\Gamma \vdash \exists x.A \quad \Gamma, A \vdash C \quad x \text{ は } \Gamma \text{ の式と } C \text{ の自由変数ではない}}{\Gamma \vdash C} \exists_e$$

```
C : Prop
X : Type
x : X
A : X -> Prop
At : exists t : X,
      A t
=====
C
```

→destruct At as [t At]→

```
C : Prop
X : Type
x : X
A : X -> Prop
t : X
At : A t
=====
C
```


どうやって存在記号が定義されている？

存在記号はペアとして実装されている:

```
Inductive ex (A : Type) (P : A -> Prop) : Prop :=  
| ex_intro : forall x : A, P x -> exists x, P x
```

↑↑⏟
witness証明= ex P
(witness はない, 証明もない)

- ▶ `exists x, P x` は `ex P` のための記法である
- ▶ タクティク `exists t` は構成子のただの適用である:
`apply (ex_intro _ t)`

同値関係の導入

- ▶ タクティク **reflexivity** は同値関係が反射関係であることを表現する

$$\frac{}{\Gamma \vdash t = t} =_i$$

```
X : Type
t : X
=====
t = t
```

→**reflexivity**→ No more subgoals.

同値関係の除去


- ▶ 同値関係の除去は書き換えとして実装されている

$$\frac{\Gamma \vdash A[x := t] \quad \Gamma \vdash t = u}{\Gamma \vdash A[x := u]} =_e$$

```
X : Type
t, u : X
A : X -> Prop
At : A t
tu : t = u
=====
A u
```

$\rightarrow \text{rewrite } <-tu^{11} \rightarrow$

```
X : Type
t, u : X
A : X -> Prop
At : A t
tu : t = u
=====
A t
```

¹¹逆向き: `rewrite ->tu` あるいは `rewrite tu`(もう少し短い) 

どうやって同値関係が実装されている？

CoQ では同値関係は `index` を持つ帰納的型として定義されている:

```
Inductive eq (A : Type) (x : A) : A -> Prop :=  
  | eq_refl : eq A x x
```

Diagram illustrating the structure of the `eq` inductive type:

- `eq` is an inductive type with one constructor, `eq_refl`.
- The parameter `A` is of type `Type`.
- The parameter `x` is of type `A`.
- The result type is `Prop`.
- The constructor `eq_refl` takes two arguments: `A` and `x`.
- The constructor `eq_refl` is annotated with `family index argument argument`.

- ▶ `x = y` は `eq _ x y` のための記法である
- ▶ `reflexivity` = `apply eq_refl`

rewriteは何をする?

帰納法の一例

- ▶ 帰納的型を定義する際, CoQ が帰納法 (とその証明!) を生成する:

```
eq_ind : forall (A : Type) (x : A) (P : A -> Prop),  
  P x -> forall y : A, x = y -> P y
```

- ▶ eq_ind は, 基本的に, 任意の P に対して, P x を P y に変換する (**Leibniz** 同値関係と言う)
- ▶ 実は, 全スライドの **rewrite** <-tu は次のタクティクと同じ:

```
apply (eq_ind _ _ At _ tu).
```

- ▶ 結局, 今回の重要なタクティクは少ない: **intros/revert**, **apply**, **destruct ... as [...]** (, **clear**)

アウトライン

CoQ の概要

CoQ での命題論理

公理と含意

論理積と論理和

矛盾と否定

CoQ での述語論理

補足

演習問題

- ▶ 論理の授業 ([Pie16], [DNR03]) の例を形式化する
- ▶ 帰納的型を使わない (つまり, CIC より CoC を使う) 論理結合子の形式化を検討する
- ▶ 本スライドを演習問題:
<https://github.com/affeldt/ssrcoq-chiba2017>

本スライドのタクティクのまとめ

`apply`, 11, 25, 30, 38, 42, 46, 49

`clear`, 43

`cut`, 11

`destruct`, 25, 28, 40, 48

`exact`, 9, 23, 38, 46

`intros`, 37

`exists`, 47, 49

`generalize`, 10, 45

`intros`, 10, 42, 45

`left`, 27, 30

`reflexivity`, 50

`revert`, 10, 45

`rewrite`, 51

`right`, 27, 30

`split`, 22, 25

参考文献 I



Sandrine Blazy and Xavier Leroy, *Mechanized semantics for the Clight subset of the C language*, J. Autom. Reasoning **43** (2009), no. 3, 263–288.



The Coq Development Team, *The Coq proof assistant reference manual*, INRIA, 2016, Version 8.5.



Thierry Coquand and Gérard Huet, *A theory of constructions (preliminary version)*, International Symposium on Semantics of Data Types, Sophia-Antipolis, 1984, Jun. 1984.



———, *Constructions: A higher order proof system for mechanizing mathematics*, Proceedings of the European Conference on Computer Algebra, Linz, Austria EUROCAL 85, April 1–3, 1985, Linz, Austria, vol. 1 (Invited Lectures), Apr. 1985, pp. 151–184.



———, *Concepts mathématiques et informatiques formalisés dans le calcul des constructions*, Tech. Report 515, INRIA Rocquencourt, Apr. 1986.



———, *The calculus of constructions*, Information and Computation **76** (1988), 95–120.



Thierry Coquand and Christine Paulin, *Inductively defined types*, Proceedings of the International Conference on Computer Logic (COLOG-88), Tallinn, USSR, December 1988, Lecture Notes in Computer Science, vol. 417, Springer, 1990, pp. 50–66.



René David, Karim Nour, and Christophe Raffalli, *Introduction à la logique*, 2ème ed., Dunod, 2003.



Georges Gonthier, Andrea Asperti, Jeremy Avigad, Yves Bertot, Cyril Cohen, François Garillot, Stéphane Le Roux, Assia Mahboubi, Russell O'Connor, Sidi Ould Biha, Ioana Pasca, Laurence Rideau, Alexey Solovyev, Enrico Tassi, and Laurent Théry, *A machine-checked proof of the odd order theorem*, Proceedings of the 4th International Conference on Interactive Theorem Proving, ITP 2013, Rennes, France, July 22–26, 2013, Lecture Notes in Computer Science, vol. 7998, Springer, 2013, pp. 163–179.

参考文献 II



Georges Gonthier, *A computer-checked proof of the four colour theorem*, Tech. report, Microsoft Research, Cambridge, 2005, Available at:
<http://research.microsoft.com/en-us/um/people/gonthier/4colproof.pdf>. Last access:
2014/08/04.



———, *Formal proof—the four-color theorem*, Notices of the American Mathematical Society **55** (2008), no. 11, 1382–1393.



Xavier Leroy, *A formally verified compiler back-end*, J. Autom. Reasoning **43** (2009), no. 4, 363–446.



Thomas Pietrzak, *Logique—logique propositionnelle—*, Licence Informatique, Université de Lille 1 Sciences et Technologies, 2016, <http://www.thomaspietrzak.com/download.php?f=CoursLogique0.pdf>.



Christine Paulin-Mohring, *Inductive definitions in the sytem coq rules and properties*, Tech. Report 92–49, LIP, École Normales Supérieure de Lyon, Dec. 1992.